Dynamics of Machines Week 12 – Exercise

Wheel Balancing



The $\vec{F}_A$ bearing force and its α angle between the vector and the x axis as well as $\vec{F}_B$ bearing force and its β angle between the vector and the x axis are measured before balancing. The α and β angles are in the xy plane.

Determine the m_1 and m_2 masses that cause the wheel unbalance.

$$m_1 = \frac{1}{R_1\omega^2}\sqrt{\left[\frac{l_2 F_A}{(l_2-l_1)}\cos\alpha + \frac{(l_2-l_{AB})}{(l_2-l_1)}F_B\cos\beta\right]^2 + \left[\frac{l_2 F_A}{(l_2-l_1)}\sin\alpha + \frac{(l_2-l_{AB})}{(l_2-l_1)}F_B\sin\beta\right]^2}$$

$$m_2 = \frac{1}{R_2\omega^2}\sqrt{\left[\frac{l_1 F_A}{(l_2-l_1)}\cos\alpha + \frac{(l_1-l_{AB})}{(l_2-l_1)}F_B\cos\beta\right]^2 + \left[\frac{l_1 F_A}{(l_2-l_1)}\sin\alpha + \frac{(l_1-l_{AB})}{(l_2-l_1)}F_B\sin\beta\right]^2}$$

Determine the phase angle of m_1 and m_2 masses that cause the wheel unbalance.

Phase angle of m_1 unbalanced mass, γ:

$$\gamma = arctg\left(\frac{\frac{l_2 F_A}{(l_2-l_1)}\sin\alpha + \frac{(l_2-l_{AB})}{(l_2-l_1)}F_B\sin\beta}{\frac{l_2 F_A}{(l_2-l_1)}\cos\alpha + \frac{(l_2-l_{AB})}{(l_2-l_1)}F_B\cos\beta}\right)$$

Phase angle of m_2 unbalanced mass, δ :

$$\delta = arctg\left(\frac{\frac{l_1 F_A}{(l_2-l_1)}\sin\alpha + \frac{(l_1-l_{AB})}{(l_2-l_1)}F_B \sin\beta}{\frac{l_1 F_A}{(l_2-l_1)}\cos\alpha + \frac{(l_1-l_{AB})}{(l_2-l_1)}F_B \cos\beta}\right)$$

The unbalanced mass forces can be determined from the measured bearing forces that caused by the unbalanced m_1, m_2 masses and their phase angles γ, δ can be determined. The wheel can be balanced with same sized m_1^*, m_2^* additional masses where they are positioned on the opposite side of diameters so the phase angles of the additional masses is $\gamma + \pi$, $\delta + \pi$.






```
// week 12 – wheel balancing
clear;
usecanvas(%F)

// Input variables:
n=60;      // [revs/min]

// Measured bearing forces and theirs angles before balancing
F_A=100;  Alfa=2.4;          // support force [N] and its angle [rad] in point A
F_B=250;  Beta=0.2;          // support force [N] and its angle [rad] in point B

L_AB=0.5;          // distance between A and B bearings [m]
L_1=0.6;           // distance between m_1 ballast mass and the A bearing [m]
L_2=0.78;          // distance between m_2 ballast mass and the A bearing [m]
R=0.2;             // radius of m_1 and m_2 ballast masses [m]

omega=(2*%pi*n)/60;        // working angular velocity [rad/s]
```

```
// az m1-es tömeghez tartozó tömegerő meghatározása
F1X=-L_2*F_A*cos(Alfa)/(L_2-L_1)-(L_2-L_AB)*F_B*cos(Beta)/(L_2-L_1);
F1Y=-L_2*F_A*sin(Alfa)/(L_2-L_1)-(L_2-L_AB)*F_B*sin(Beta)/(L_2-L_1);
F1=sqrt(F1X^2+F1Y^2);

// m_1 additional mass properties:
m_1=F1/(R*omega^2);       // m_1 additional mass (kg)
Gamma=atan(F1Y/F1X);      // phase angle of m1 mass (rad)
Gamma_m1=Gamma+%pi;  // phase angle of m_1 additional mass (rad)

// Display results in Console window
m_1                       // m_1 additional mass (kg)
Gamma_m1*(180/%pi)        // mounting angle of m_1 additional mass onto the wheel rim (deg)

// az m2-es tömeghez tartozó tömegerő meghatározása
F2X=L_1*F_A*cos(Alfa)/(L_2-L_1)+(L_1-L_AB)*F_B*cos(Beta)/(L_2-L_1);
F2Y=L_1*F_A*sin(Alfa)/(L_2-L_1)+(L_1-L_AB)*F_B*sin(Beta)/(L_2-L_1);
F2=sqrt(F2X^2+F2Y^2);

// m_2 additional mass properties:
m_2=F2/(R*omega^2);       // m_2 additional mass (kg)
Delta=atan(F2Y/F2X);      // phase angle of m2 mass (rad)
Delta_m2=Delta+%pi;       // phase angle of m_2 additional mass (rad)

// Display results in Console window
m_2                       // m_2 additional mass (kg)
Delta_m2*(180/%pi)        // mounting angle of m_2 additional mass onto the wheel rim (deg)

// Plotting ----------------------------------------
aa=gca();
aa.isoview='on';

plotframe([-1.2*R, -1.2*R, 1.2*R, 1.2*R])

// Draw a circle with R radius
fi=0:0.01:2*%pi;
xr=R*cos(fi);
yr=R*sin(fi);
plot(xr,yr);

x_m1=R*cos(Gamma_m1);             // x coordinate of m1 additional mass
y_m1=R*sin(Gamma_m1);             // y coordinate of m1 additional mass
x_m2=R*cos(Delta_m2);             // x coordinate of m2 additional mass
y_m2=R*sin(Delta_m2);             // y coordinate of m2 additional mass
plot(x_m1,y_m1,'ro');
plot(x_m2,y_m2,'ro');

fac=1.1;
mx_m1=fac*R*cos(Gamma_m1);        // x coordinate of m1 additional mass label
my_m1=fac*R*sin(Gamma_m1);        // y coordinate of m1 additional mass label
mx_m2=fac*R*cos(Delta_m2);        // x coordinate of m2 additional mass label
my_m2=fac*R*sin(Delta_m2);        // y coordinate of m2 additional mass label
xstring(mx_m1,my_m1,["m_1"]);
xstring(mx_m2,my_m2,["m_2"]);
```

```
x_FA=R*cos(Alfa);                // x coordinate of FA support force
y_FA=R*sin(Alfa);                // y coordinate of FA support force
x_FB=R*cos(Beta);                // x coordinate of FB support force
y_FB=R*sin(Beta);                // y coordinate of FB support force
plot(x_FA,y_FA,'+');
plot(x_FB,y_FB,'+');

fac=1.1;
fx_FA=fac*R*cos(Alfa);           // x coordinate of FA support force label
fy_FA=fac*R*sin(Alfa);           // y coordinate of FA support force label
fx_FB=fac*R*cos(Beta);           // x coordinate of FB support force label
fy_FB=fac*R*sin(Beta);           // y coordinate of FB support force label
xstring(fx_FA,fy_FA,["F_A"]);
xstring(fx_FB,fy_FB,["F_B"]);
```

```
--> // Display results in Console window
--> m_1                              // m_1 additional mass (kg)
 m_1  =

   47.501173
--> Gamma_m1*(180/%pi)   // mounting angle of m_1 additional mass
 ans  =

   260.54679
--> // Display results in Console window
--> m_2                              // m_2 additional mass (kg)
 m_2  =

   34.894813
--> Delta_m2*(180/%pi)   // mounting angle of m_2 additional mass
 ans  =

   113.45802
```


m_2
F_A
F_B
m_1
0.2
0.15
0.1
0.05
0
-0.05
-0.1
-0.15
-0.2
-0.2
-0.15
-0.1
-0.05
0
0.05
0.1
0.15
0.2